
******************************************
**** Study 2 - Ognyanova et al. 2025 *****
*********** Analyses do file *************
******************************************

* Changing directory
**<<<ADD DIRECTORY>>>

* Loading the dataset data (.dta)
insheet using "Study 2 data.csv", clear


* Respodents ID
gen id=_n

******************
** Demographics **
******************

* Gender
gen gender2=1 if gender=="Female"
replace gender2=0 if gender=="Male"
label define gender_lab 0 "Male" 1 "Female"
label values gender2 gender_lab
tab gender gender2
drop gender 
rename gender2 gender
label var gender "Gender (female=1)"
fre gender


* Age group
gen age_group=1 if age_cat_6=="18 to 24"
replace age_group=2 if age_cat_6=="25 to 34"
replace age_group=3 if age_cat_6=="35 to 44"
replace age_group=4 if age_cat_6=="45 to 54"
replace age_group=5 if age_cat_6=="55 to 64"
replace age_group=6 if age_cat_6=="65 and over"
label define age_g_lab 1 "18-24" 2 "25-34" 3 "35-44" 4 "45-54" 5 "55-64" 6 "65+"
label value age_group age_g_lab
tab age_cat_6 age_group
label var age_group "Age groups"
fre age_group

* White respondent
gen white=1 if race=="White"
replace white=0 if race!="White"
label define white_lab 0 "Non-White" 1 "White"
label values white white_lab
tab race white
label var white "White respondent"
fre white

* College education
rename education_cat educ
gen college_educ=1 if educ=="College Degree" | educ=="Graduate Degree" | ///
					  educ=="Some College"
replace college_educ=0 if educ=="High School Graduate" | ///
					      educ=="Some High School or Less" 
label define coll_lab 0 "No college education" 1 "College education"
label values college_educ coll_lab
tab educ college_educ
label var college_educ "College education"
fre college_educ

* Income
gen income=income_cat_10  
label var income "Income (10 categories)"

* Republican respondent
gen rep=1 if party=="Republican"
replace rep=0 if party=="Democrat"
label define rep_lab 0 "Democrat" 1 "Republican"
label values rep rep_lab
tab party rep, miss
label var rep "Republican respondent"
fre rep

* The Conspiracy thinking scale
gen CT_scale=consp 
replace CT_scale=(CT_scale-1)/4
label var CT_scale "The Conspiracy thinking scale"
sum CT_scale

* The "pro-Republican" Trummp assassination conspiracy theory
tab assass_fn_bel_1
gen CON_PRO_REP=1 if assass_fn_bel_1=="5"
replace CON_PRO_REP=2 if assass_fn_bel_1=="4"
replace CON_PRO_REP=3 if assass_fn_bel_1=="3"
replace CON_PRO_REP=4 if assass_fn_bel_1=="2"
replace CON_PRO_REP=5 if assass_fn_bel_1=="1"
replace CON_PRO_REP=. if assass_fn_bel_1=="NA"
label define likely_lab 1 "Very unlikely" 2 "Somewhat unlikely" ///
						3 "Neither likely nor unlikely" 4 "Somewhat likely" ///
						5 "Very likely"
label values CON_PRO_REP likely_lab
label var CON_PRO_REP "The 'pro-Republican' assassination CT, 5-point"
tab assass_fn_bel_1 CON_PRO_REP, miss
fre CON_PRO_REP


* The "pro-Democrat" Trump assassination conspiracy theory
tab assass_fn_bel_2
gen CON_PRO_DEM=1 if assass_fn_bel_2=="5"
replace CON_PRO_DEM=2 if assass_fn_bel_2=="4"
replace CON_PRO_DEM=3 if assass_fn_bel_2=="3"
replace CON_PRO_DEM=4 if assass_fn_bel_2=="2"
replace CON_PRO_DEM=5 if assass_fn_bel_2=="1"
replace CON_PRO_DEM=. if assass_fn_bel_2=="NA"
label values CON_PRO_DEM likely_lab
label var CON_PRO_DEM "The 'pro-Democrat' assassination CT, 5-point"
tab assass_fn_bel_2 CON_PRO_DEM, miss
fre CON_PRO_DEM


*** Beliefs in political conspiracy theories ***
tab1 CON_PRO_REP CON_PRO_DEM
recode CON_PRO_REP (1/3=0)(4/5=1), gen(CTI_PRO_REP_2)
tab CON_PRO_REP CTI_PRO_REP_2
recode CON_PRO_DEM (1/3=0)(4/5=1)(6=0), gen(CTI_PRO_DEM_2)
tab CON_PRO_DEM CTI_PRO_DEM_2
label define ct_agree 0 "Disagree/Neutral/DK" 1 "Agree"
label values CTI_PRO_REP_2 ct_agree
label values CTI_PRO_DEM_2 ct_agree
tab1 CTI_PRO_REP_2 CTI_PRO_DEM_2 


** RESHAPE of the data - from Wide to Long
* Conspiracy theory number
*reshape long CON_, i(id) j(CTI_num) string
reshape long CTI_, i(id) j(CTI_string) string
order CTI_string, after(CTI_)
fre CTI_string
gen CTI_num=1 if CTI_string=="PRO_REP_2"
replace CTI_num=2 if CTI_string=="PRO_DEM_2"
label var CTI_num "Conspiracy theory number"
label define CTI_num_lab 1 "PRO-REPUBLICAN CT" 2 "PRO-DEMOCRAT CT" 
label values CTI_num CTI_num_lab
fre CTI_num

* The DV: Belief in the CT
rename CTI_ belief_CT
tab belief_CT
tab belief_CT CTI_num if rep==1, chi col
tab belief_CT CTI_num if rep==0, chi col

* "Congenial bloc" dummy
gen congenial_bloc=1 if CTI_num==1 & rep==1
replace congenial_bloc=0 if CTI_num==1 & rep==0
replace congenial_bloc=1 if CTI_num==2 & rep==0
replace congenial_bloc=0 if CTI_num==2 & rep==1

label var congenial_bloc "Congenial bloc"
tab congenial_bloc CTI_num if rep==1
tab congenial_bloc CTI_num if rep==0


* Combined "Political" CT analyses
reg belief_CT c.CT_scale##i.congenial_bloc i.gender i.age_group i.white i.college_educ income i.CTI_num, cluster(id)
outreg2 using Table_1.doc, append se dec(3) alpha (.001, .01, .05) ///
symbol (***, **, *) keep(c.CT_scale##i.congenial_bloc) ///
ctitle ("Model 2 - July 2024 (Ognyanova et al. 2024)") ///
addtext(CT Fixed-effects, "YES", Individual-level controls, "YES") ///
title ("Table 1. Predicting belief in political conspiracy theories - US studies") 

* Post-hoc power analysis:
retrodesign .3291896, se(.0769264) alpha(0.05) /*Power=.990*/


* Calculating the coef. of the CT scale among those in the "congenial bloc"
gen inter=CT_scale*congenial_bloc
reg belief_CT CT_scale congenial_bloc inter i.gender i.age_group i.white i.college_educ income i.CTI_num, cluster(id)
lincom CT_scale+inter
drop inter


** Predicting specific "political" conspiracy theories **

* The "pro-Republican" Trump assassination conspiracy CT 
reg belief_CT c.CT_scale##i.rep i.gender i.age_group i.white i.college_educ income if CTI_num==1, r /*p<.001, in the expected direction*/

* The "pro-Democrat" Trump assassination CT 
reg belief_CT c.CT_scale##i.rep i.gender i.age_group i.white i.college_educ income if CTI_num==2, r /*p=.017, in the expected direction*/

/* 
In total, 2/2 statistically significant (p<.05; 1 "for" Reps. + 1 "for" Dems.) 
*/


************************************
** Full results - Online Appendix **
************************************

* Table C1 - "Political" CT analyses in the US
reg belief_CT c.CT_scale##i.congenial_bloc i.gender i.age_group i.white i.college_educ income i.CTI_num, cluster(id)
outreg2 using TableC1.doc, append se dec(3) alpha (.001, .01, .05) ///
symbol (***, **, *) drop(i.CTI_num) ///
ctitle ("Model 2 - Study 2") addtext(CT Fixed-effects, "YES") ///
title ("Table C1. Predicting belief in political conspiracy theories, US studies – Full results") 


*********************************************
** Dropping all controls - Online Appendix **
*********************************************

* Table D1 - "Political" CT analyses in the US
reg belief_CT c.CT_scale##i.congenial_bloc i.CTI_num, cluster(id)
outreg2 using TableD1.doc, append se dec(3) alpha (.001, .01, .05) ///
symbol (***, **, *) keep(c.CT_scale##i.congenial_bloc) ///
ctitle ("Model 2 - Study 2") addtext(CT Fixed-effects, "YES") ///
title ("Table D1. Predicting belief in political conspiracy theories, US studies – No controls") 

* Calculating the coef. of the CT scale among those in the "congenial bloc"
gen inter=CT_scale*congenial_bloc
reg belief_CT CT_scale congenial_bloc inte i.CTI_num, cluster(id)
lincom CT_scale+inter
drop inter



****************************
** Descriptive statistics **
****************************

* Changing directory
**<<<ADD DIRECTORY>>>

* Loading the dataset data (.dta)
insheet using "Study 2 data.csv", clear

* Gender
gen gender2=1 if gender=="Female"
replace gender2=0 if gender=="Male"
*label define gender_lab 0 "Male" 1 "Female"
label values gender2 gender_lab
tab gender gender2
drop gender 
rename gender2 gender
label var gender "Gender (female=1)"
fre gender

* College education
rename education_cat educ
gen college_educ=1 if educ=="College Degree" | educ=="Graduate Degree" | ///
					  educ=="Some College"
replace college_educ=0 if educ=="High School Graduate" | ///
					      educ=="Some High School or Less" 
*label define coll_lab 0 "No college education" 1 "College education"
label values college_educ coll_lab
tab educ college_educ
label var college_educ "College education"
fre college_educ

* White respondent
gen white=1 if race=="White"
replace white=0 if race!="White"
*label define white_lab 0 "Non-White" 1 "White"
label values white white_lab
tab race white
label var white "White respondent"
fre white

* Republican respondent
gen rep=1 if party=="Republican"
replace rep=0 if party=="Democrat"
*label define rep_lab 0 "Democrat" 1 "Republican"
label values rep rep_lab
tab party rep, miss
label var rep "Republican respondent"
fre rep

